

「EQUATRAN-M」技術計算用連立方程式解法言語

(8)

EQUATRAN-M のより広い分野への応用

山 田 明*

大 村 憲 一**


宮 原 昱 中***

今回は EQUATRAN-M のより広い分野への応用として電気・電子分野，計測・制御分野，機械工学分野の例について解説する。ケミカルエンジニアにはなじみのない例が多いが，化学工業の周辺技術を支える各分野の技術者にとっても EQUATRAN-M が活用できることを示す。

1. 交流ブリッジ回路の解析

電気・電子の分野では回路網解析を行うことにより各素子に流れる電流や端子間の電圧を決定する問題がある。ここでは交流ブリッジ回路の解析を例にして EQUATRAN-M で回路解析がどのようにできるかを示す。交流回路では複素数を取り扱う必要があるが，これも簡単に表現できることを示そう。

<例題 8.1：交流ブリッジ回路の解析>¹⁾



左図に示すような交流ブリッジ回路において各枝路に流れる電流を求める。各枝路電流は複素数となるので，直角座標表示と極座標表示の両方で求めよ。こ

こで回路の定数および電圧は $Z_1 = -j10\Omega$ ，

$Z_2 = j5\Omega$ ， $Z_3 = 40\Omega$ ， $Z_4 = -j10\Omega$ ，
 $Z_5 = j5\Omega$ ， $E = 150V$ とする。

解) 回路解析法には枝路電流法，閉路電流法，節点解析法の3つの方法があるが，ここでは枝路電流法による方法で解いてみよう。枝路電流法は Kirchhoff の第1法則と第2法則を利用して，枝路電流を未知数として電流分布を求める方法である。第1法則により，各枝路電流 I_i について，

$$I_0 = I_1 + I_2 \quad (8-1)$$

$$I_1 = I_3 + I_5 \quad (8-2)$$

$$I_4 = I_2 + I_5 \quad (8-3)$$

各枝路の電圧降下 $V_i = Z_i I_i$ として，第2法則により，各部の電圧について，

$$E = V_2 + V_4 \quad (8-4)$$

$$V_2 = V_1 + V_5 \quad (8-5)$$

$$V_3 = V_4 + V_5 \quad (8-6)$$

が成立する。ところでそれぞれの変数は全て複素数であるので，演算について特別な配慮が必要である。直角座標表示の複素数 $A = a_1 + j a_2$ ， $B = b_1 + j b_2$ についての加算および乗算は

$$A + B = (a_1 + b_1) + j(a_2 + b_2)$$

$$A \times B = (a_1 b_1 - a_2 b_2) + j(a_1 b_2 + a_2 b_1)$$

で表わすことができる。また複素数 A の極座標表示は絶対値を r ，位相角を θ とすれば，

$$A = r e^{j\theta} = r \cos \theta + j r \sin \theta$$

となるので， a_1 ， a_2 と r ， θ の関係は，

$$r = \sqrt{a_1^2 + a_2^2} \quad (8-7)$$

* Akira Yamada 三井東圧化学(株)システム部 主務

** Kenichi Ohmura 三井東圧化学(株)システム部 主査

*** Koreatsu Miyahara 三井東圧化学(株)システム部 次長

表 8-1 例題 8.1 の解

```

1: /* 交流ブリッジ回路の解析 (枝路電流法) */
2: LOCAL N = 5 /* 枝路数 */
3: VAR Z(N,2) /* 枝路インピーダンス [ohm] */
4: VAR I(N,2) /* 枝路電流 [A] */
5: VAR V(N,2) /* 枝路電圧降下 [V] */
6: VAR E(2) /* 印加電圧降下 [V] */
7: VAR IO(2) /* 枝路電流 [A] */
8: VAR i(N,2) /* 枝路電流 (極座標表示) [A] */
9: /* キルヒホッフの第一法則 */
10: IO = I(1) + I(2)
11: I(1) = I(3) + I(5)
12: I(4) = I(2) + I(5)
13: /* キルヒホッフの第二法則 */
14: E = V(2) + V(4)
15: V(2) = V(1) + V(5)
16: V(3) = V(4) + V(5)
17: /* 電圧降下計算 V = Z * I */
18: V(,1) = Z(,1)*I(,1) - Z(,2)*I(,2)
19: V(,2) = Z(,1)*I(,2) + Z(,2)*I(,1)
20: /* 電流の絶対値および位相角への変換 (極座標表示) */
21: i(,1) = sqrt(I(,1)^2 + I(,2)^2)
22: i(,2) = atan(I(,2)/I(,1))*180/3.1416
23: WHEN (i(,1) > 0)
24:   = atan(I(,2)/I(,1))*180/3.1416
25:   + 180 WHEN (I(,1) < 0 & I(,2) >= 0)
26:   = atan(I(,2)/I(,1))*180/3.1416
27:   - 180 WHEN (I(,1) < 0 & I(,2) < 0)
28:   = 0 WHEN (I(,1) == 0 & I(,2) == 0)
29:   = 90 WHEN (I(,1) == 0 & I(,2) > 0)
30:   = -90 WHEN (I(,1) == 0 & I(,2) < 0)
31:
32: INPUT E, Z
33: OUTPUT I, i
34:
35: [ 入力値(=) または 初期値(θ) ]
E (2) = 150 0
Z (2*5) = 0 -10
         0 5
         40 0
         0 -10
         0 5

[ 計算結果 ]
I =
( 1) ( 2)
1) -7.2 24.6
2) 9.6 -22.8
3) -2.4 -1.8
4) 4.8 3.6
5) -4.8 26.4
i =
( 1) ( 2)
1) 25.632011 106.314025
2) 24.738634 -67.166189
3) 3 -143.130189
4) 6 36.869811
5) 26.832816 100.305033

```

$$\theta = \tan^{-1}(a_2/a_1) \quad (8-8)$$

となる。表 8-1 にこの問題に対するリストと計算結果を示す。全ての変数を $N \times 2$ の配列でとっているが、 N は枝路の数に対応し、 2 は直角座標表示での実数部と虚数部に対応させている。ただし 8 行の小文字の変数 i は I の極座標表示であり、絶対値 r と位相角 θ に対応させている。21~32 行で直角座標から極座標への変換をしているが、位相角については複素数平面上の象限による場合わけを行っている。18~32 行は回路網の形によらず、いつもこのように書けばよい。

ここでは交流回路の例を示したが、直流回路の場合は複素数を考える必要がなく配列を 1 次元で考えればよい。またさらに複雑な回路網（鉄心を含むコイル、トランジスタ、ダイオードなどの非線形要素からなる）についても、表や条件文を導

入することにより EQUATRAN-M で解くことができるだろう。

2. オリフィス設計計算

計測分野では管内を流れる流体の流量計測のためにオリフィス板がよく利用される。このオリフィス設計の基本的な計算法は日本工業規格 (Z 8762) で規定されているが、2 次元数表と繰り返し計算が必要なためかなり面倒である。ここでは EQUATRAN-M の 2 次元テーブルを利用することにより、簡単に設計計算ができることを示す。

<例題 8.2: オリフィス設計計算>

次の条件でガスの流量計測を行うためのコーナタップオリフィスの絞り孔径 d [mm] を求めよ。

最大流量 $W = 30 \text{ kg/h}$, 流体圧力 $P_1 = 4 \text{ Kg/cm}^2 \text{G}$, 最大差圧 $\Delta P = 1,000 \text{ mmH}_2\text{O}$, ガスの密度 $\gamma_1 = 7.44 \text{ kg/m}^3$, 粘度 $\mu = 0.014 \text{ cp}$, 比熱比 $\kappa = 1.4$, 上流側管径 $D = 19.4 \text{ mm}$, 管のあらさ係数 $k = 0.08 \text{ mm}$

解) 日本工業規格のコーナタップオリフィスの設計規定によると、基本的な関係式は

$$Q = \alpha \epsilon \frac{\pi}{4} d^2 \sqrt{\frac{2 \Delta P}{\gamma_1}} \quad (8-9)$$

$$W = \gamma_1 Q \quad (8-10)$$

$$\alpha = \alpha_0 r_{Ra} \quad (8-11)$$

$$r_{Ra} = \begin{cases} (r_0 - 1) \left(\frac{\log_{10} R_d}{6} \right)^2 + 1 & (R_d < 10^6) \\ r_0 & (R_d \geq 10^6) \end{cases} \quad (8-12)$$

$$\epsilon = 1 - (0.3707 + 0.3184 \beta^4) \times \left(1 - \left(\frac{P_2}{P_1} \right)^{1/\kappa} \right)^{0.935} \quad (8-13)$$

$$R_d = \frac{u D \gamma_1}{\mu} \quad (8-14)$$

$$\beta = \frac{d}{D} \quad (0.22 < \beta < 0.8) \quad (8-15)$$

で与えられている。上式中の α_0 はレイノルズ数 R_d と絞り直径比 β の関数として、 r_0 はレイノルズ数 R_d と管内壁の相対あらさ k/D の関数として 2 次元の表で与えられている。EQUATRAN-M では 2 次元の表を簡単にとり扱うことができるので、 α_0 や r_0 を関数近似することなく自然な形で

表 8-2 例題 8.2 の解

```

1: /*      コーナタップ オリフィス 設計計算      */
2: /*      日本工業規格      Z8762-1969      */
3: /*      流量係数 テーブル      */
4: VAR x1(8),x2(11),x(11,8)
5: TABLE x = alph0(x2,x1) /* x1=Rd, x2=b^4 */
6: x1 = (5E3, 1E4, 2E4, 3E4, 5E4, 1E5, 1E6, 1E7)
7: x2 = (0.0025, 0.005, 0.01, 0.04, 0.10, 0.15, ...
8:      0.20, 0.25, 0.30, 0.35, 0.40)
9: x = ..
10: (0.603,0.600,0.599,0.599,0.598,0.598,0.598,0.597)..
11: (0.606,0.602,0.602,0.602,0.601,0.601,0.600,0.599)..
12: (0.611,0.606,0.605,0.604,0.603,0.603,0.602,0.602)..
13: (0.634,0.626,0.621,0.618,0.617,0.616,0.615,0.614)..
14: (0.658,0.650,0.645,0.642,0.640,0.637,0.636)..
15: (0.684,0.673,0.668,0.663,0.659,0.656,0.655)..
16: (0.710,0.696,0.689,0.688,0.679,0.675,0.674)..
17: (0.737,0.719,0.712,0.705,0.699,0.695,0.693)..
18: (0.763,0.743,0.735,0.727,0.720,0.715,0.713)..
19: (0.792,0.770,0.760,0.750,0.743,0.738,0.733)..
20: (0.820,0.797,0.786,0.777,0.767,0.757,0.756)
21: /*      管内壁あらさ補正用 r0 テーブル      */
22: VAR z1(8),z2(7),z(7,8)
23: TABLE z = r0(z2,z1) /* z1=D/k, z2=b^2 */
24: z1 = (400, 800, 1200,1600, 2000, 2400, 2800, 3200)
25: z2 = (0.1, 0.2, 0.3, 0.4, 0.5, 0.6, 0.64)
26: z = ..
27: (1.002,1.000,1.000,1.000,1.000,1.000,1.000,1.000)..
28: (1.003,1.002,1.001,1.000,1.000,1.000,1.000,1.000)..
29: (1.006,1.004,1.002,1.001,1.000,1.000,1.000,1.000)..
30: (1.009,1.006,1.004,1.002,1.001,1.000,1.000,1.000)..
31: (1.014,1.009,1.006,1.004,1.002,1.001,1.000,1.000)..
32: (1.020,1.013,1.009,1.006,1.003,1.002,1.000,1.000)..
33: (1.024,1.016,1.011,1.007,1.004,1.002,1.002,1.000)
34:
35: /*      Q      体 流 量      [m3/h]      */
36: /*      W      重量流量      [kg/h]      */
37: /*      P1     流体圧力      [kg/cm2G]      */
38: /*      delP   差圧      [mmH2O]      */
39: /*      gam1   流体密度      [kg/m3]      */
40: /*      mu     流体粘度      [cp]      */
41: /*      kappa  比熱比      [-]      */
42: /*      D      上流側管径      [mm]      */
43: /*      k      管のあらさ      [mm]      */
44: /*      d      絞り孔径      [mm]      */
45: /*      b      絞り直径比      [-]      */
46: /*      alph   流量係数      [-]      */
47: /*      eps    膨張補正係数      [-]      */
48: /*      Rd     レイノルズ数      [-]      */
49: /*      u      管内流速      [m/s]      */
50:
51: eq:Q = 0.01252*alph*eps*d^2*SQRT(delP/gam1)
52: W = Q * gam1
53: alph = alph0(b^4, Rd)*rRd
54: rRd = ( r0(b^2, D/k) - 1 ) * ( LOG10(Rd)/6 )^2 + 1
55:      WHEN Rd < 1E6
56:      = r0(b^2, D/k) WHEN Rd >= 1E6
57: eps = 1 - (0.3707 + 0.3184*b^4)
58:      * (1 - ((P2+1.033)/(P1+1.033))^(1/kappa))
59:      * 0.935
60: Rd = u*D*gam1/mu
61: b = d/D
62: u = Q*1E6*4/3.1416/D^2/3600
63: delP = (P1 - P2)*10000
64:
65: INPUT W,P1,delP,gam1,mu,kappa,D,k
66: OUTPUT d,b,alph,eps,Rd
67:
68: RESET b #0.5 [0.22, 0.8] BY eq
( 入力値(=) または 初期値(≠) )
V      = 30
P1     = 4
delP   = 1000
gam1   = 7.44
mu     = 0.014
kappa  = 1.4
D      = 19.4
k      = 0.08
b      = 0.5
[ 計算結果 ]
d      = 6.789521
b      = 0.349975
alph   = 0.606919
eps    = 0.992055
Rd     = 39065.923152

```

問題を解くことができる(2次元の表を関数近似するのは非常に面倒である)。表8-2にこの問題に対するリストと計算結果を示す。3~20行に α_0 の表、21~33行に r_0 の表がある。(8-9)式から(8-15)式までの基本的な関係式は51~63行に記述してあるが、単位系を合わせるために各式を少し変形してある。 r_{Rd} はレイノルズ数 Rd により式が変わるが WHEN 文により自然な形で表現されている。この規格の適用範囲は絞り直径比 β が0.22か


ら0.8の範囲であるので68行に示されているように収束計算をその範囲で行うように設定している。このように EQUATRAN-M では自然な表現でオリフィス設計計算を実現できることがわかる。またすでにあるオリフィスを使用して流量レンジを変更する問題がよくあるが、このリストを少し変えるだけで対応できる。65行の delP と66行の d を入れかえて、68行の RESET 文をとってしまえば OK である。

3. 管型反応器の最適制御

現代制御理論の分野では最適制御の問題がよくとり扱われている。Pontryagin の“最大原理”は解析的に最適制御方法を求める手法として有名であるが、これを管型反応器の最適温度制御に適用すると2点境界値問題になってしまう、なかなか解くことが難しい。ここでは EQUATRAN-M でこの2点境界値問題をどのように解くかを示す。

<例題 8.3: 管型反応器の最適制御>³⁾

押し出し流れの管型反応器でつぎのような逐次反応が起こり、 B が希望する製品である。



この2つの反応はいずれも1次反応であり、次式に示すような反応モデルで表わされるものとする。

$$r_{AB} = k_1 x_1 \quad [\text{mol/l min}]$$

$$r_{BC} = k_2 x_2 \quad [\text{mol/l min}]$$

$$k_1 = k_{10} \exp\left(-\frac{E_1}{RT}\right) \quad [1/\text{min}]$$

$$k_2 = k_{20} \exp\left(-\frac{E_2}{RT}\right) \quad [1/\text{min}]$$

反応器の入口における A および B の濃度 $\alpha_1 = 0.53 \text{ mol/l}$, $\alpha_2 = 0.43 \text{ mol/l}$, 反応器の保持時間を 8 min としたとき、反応器の出口における B の濃度を最大にする最適温度 T のパターンを求めよ。

$$k_{10} = 0.535 \times 10^{11} \quad [1/\text{min}]$$

$$k_{20} = 0.461 \times 10^{18} \quad [1/\text{min}]$$

$$E_1 = 18000 \quad [\text{cal/mol}]$$

$$E_2 = 30000 \quad [\text{cal/mol}]$$

$$R = 2 \quad [\text{cal/mol K}]$$

解) この問題に対して“最大原理”を適用した解析の詳細は文献³⁾に詳しく記述してあるので、ここではその結果として導出される2点境界値問題について考えよう。文献によると上記問題の解は次の2点境界値問題を解けば求められる。

$$\frac{dx_1}{dt} = -k_{10} y^{\lambda_1} x_1 \quad (8-16)$$

$$\frac{dx_2}{dt} = k_{10} y^{\lambda_1} x_1 - k_{20} y^{\lambda_2} x_2 \quad (8-17)$$

$$\frac{dz_1}{dt} = k_{10} y^{\lambda_1} (z_1 - z_2) \quad (8-18)$$

$$\frac{dz_2}{dt} = k_{20} y^{\lambda_2} z_2 \quad (8-19)$$

$$y = \Psi \frac{x_1(z_2 - z_1)}{x_2 z_1} = \exp\left(\frac{E_1 - E_2}{RT}\right) \quad (8-20)$$

$$\Psi = \frac{k_{10} E_1}{k_{20} E_2} \quad (8-21)$$

$$\lambda_1 = \frac{E_1}{E_2 - E_1}, \quad \lambda_2 = \frac{E_2}{E_2 - E_1} \quad (8-22)$$

$$x_1(0) = \alpha_1, \quad x_2(0) = \alpha_2$$

$$z_1(8) = 0, \quad z_2(8) = 1$$

表 8-3 例題 8.3 のリスト

```

1: /* 管型反応器の最適制御 (二点境界値問題) */
2: LOCAL N=41 /* 分割数 */
3: VAR t(N), x1(N), x1D(N-1), y(N-1), TT(N-1)
4: VAR x2(N), x2D(N-1), T(N)
5: VAR z1(N), z1D(N-1)
6: VAR z2(N), z2D(N-1)
7: /* t の増分 */
8: Dt = (t(N)-t(1))/(N-1)
9: t(2:N-1) = t(1:N-2)+Dt
10: /* 差分近似 */
11: x1D = (x1(2:N)-x1(1:N-1))/Dt
12: x2D = (x2(2:N)-x2(1:N-1))/Dt
13: z1D = (z1(2:N)-z1(1:N-1))/Dt
14: z2D = (z2(2:N)-z2(1:N-1))/Dt
15: /* 微分方程式 */
16: x1D = -k10*y^L1*x1(1:N-1)
17: x2D = k10*y^L1*x1(1:N-1)-k20*y^L2*x2(1:N-1)
18: z1D = k10*y^L1*(z1(2:N)-z2(2:N))
19: z2D = k20*y^L2*z2(2:N)
20:
21: eq:y = psi*x1(1:N-1)/x2(1:N-1)
22: /* (z2(2:N)-z1(2:N))/z2(2:N) */
23: y = exp((E1-E2)/R/TT)
24: psi = k10/k20*E1/E2
25: L1 = E1/(E2-E1)
26: L2 = E2/(E2-E1)
27:
28: /* 境界条件 */
29: t(1) = 0 ; t(N) = 8
30: x1(1) = 0.53 ; z1(N) = 0
31: x2(1) = 0.43 ; z2(N) = 1
32: /* 反応定数他 */
33: k10 = 0.535E+11 ; k20 = 0.461E18
34: E1 = 18000 ; E2 = 30000 ; R = 2
35: /* 温度の外挿 */
36: T(1:N-1) = TT(1:N-1)
37: T(N) = TT(N-1)
38: /* + 2*TT(N-1)-3*TT(N-2)+TT(N-3) */
39:
40: RESET y # 2.0E-8[0, 1.0E-7] BY eq
41:
42: /* OUTPUT t,x1,x2,z1,z2,T,TT */
43: OUTPUT t(1.5.9.13.17.21.25.29.33.37.41)
44: OUTPUT x1(1.5.9.13.17.21.25.29.33.37.41)
45: OUTPUT x2(1.5.9.13.17.21.25.29.33.37.41)
46: OUTPUT z1(1.5.9.13.17.21.25.29.33.37.41)
47: OUTPUT z2(1.5.9.13.17.21.25.29.33.37.41)
48: OUTPUT T(1.5.9.13.17.21.25.29.33.37.41)

```

表 8-4 例題 8.3 の計算結果

計算結果			
t(1.5.9.13.17.21.25.29.33.37.41) =	0	8	131 2.4
1) 0	5) 0.8	9) 1.6	25) 4.8
17) 3.2	21) 4	29) 5.6	
33) 6.4	37) 7.2		
x1(1.5.9.13.17.21.25.29.33.37.41) =			
1) 0.53	5) 0.458157	9) 0.401142	13) 0.354212
17) 0.314697	21) 0.280907	25) 0.251667	29) 0.226155
33) 0.203738	37) 0.183831	41) 0.166348	
x2(1.5.9.13.17.21.25.29.33.37.41) =			
1) 0.43	5) 0.489361	9) 0.534361	13) 0.569628
17) 0.507773	21) 0.620451	25) 0.638793	29) 0.653618
33) 0.665537	37) 0.675024	41) 0.682453	
z1(1.5.9.13.17.21.25.29.33.37.41) =			
1) 0.61755	5) 0.58263	9) 0.542965	13) 0.498164
17) 0.447814	21) 0.391455	25) 0.328583	29) 0.258642
33) 0.181021	37) 0.095051	41) 0	
z2(1.5.9.13.17.21.25.29.33.37.41) =			
1) 0.825934	5) 0.8491	9) 0.869539	13) 0.888294
17) 0.905908	21) 0.922704	25) 0.938894	29) 0.954625
33) 0.970002	37) 0.985106	41) 1	
T(1.5.9.13.17.21.25.29.33.37.40) =			
1) 341.129554	5) 339.802058	9) 338.88006	13) 338.187674
17) 337.642616	21) 337.199726	25) 336.831563	29) 336.520217
33) 336.253374	37) 336.022207	40) 335.868253	




図 8-1 最適温度と濃度分布の比較

ここで z_1, z_2 は“最大原理”適用により導入された補助変数である。この問題を EQUATRAN-M で解いたリストを表 8-3 に示す。この例では微分方程式を差分近似して解くことにより、2点境界値問題を連立方程式の解法問題に変えている。7~14行は差分近似のための式であり、15~19行は(8-16)式から(8-19)式に対応した微分方程式に相当する。2行の変数 N は区間の大きさであ

表 8-5 最適温度パターンの比較

(分割数Nを変えたとき)

[K]

無次元 時間	文献3)の 結果	EQUATRAN-M の結果				
		N=11	N=21	N=31	N=41	N=61
0	340.79	342.71	341.49	341.22	341.13	341.02
0.1	339.64	340.15	339.93	339.83	339.80	339.75
0.2	338.81	338.86	338.91	338.88	338.88	338.86
0.3	338.16	338.00	338.17	338.18	338.19	338.19
0.4	337.64	337.36	337.59	337.62	337.64	337.65
0.5	337.22	336.86	337.13	337.17	337.20	337.21
0.6	336.87	336.45	336.75	336.80	336.83	336.85
0.7	336.56	336.12	336.43	336.49	336.52	336.54
0.8	336.31	335.83	331.15	336.22	336.25	336.28
0.9	336.08	335.59	335.92	335.98	336.02	336.05
1.0	335.88	335.40	335.72	335.79	335.82	335.85

り、分割数は $N-1$ である。21~26行は(8-20)式から(8-22)式に対応し、28~31行は境界条件である。40行に示されるようにこの計算は変数 y について分割数に比例した多変数収束計算を必要とする。温度は23行の関係式から決まるが、 $N-1$ までしか求められないので、35~38行で N 番目を外挿により求めた。計算結果の出力は通常42行のように書けばよいが、ここでは出力量を減らすために43~48行のように4キザミ間隔で出力するようにした。 $N=41$ とした時の計算結果が表8-4に示されているが、それぞれの境界条件が満足されていることがわかる。この結果から反応器出口における濃度は A が 0.1663mol/l 、 B が 0.6825mol/l となり、そのときの最適温度 T は入口で 341.13K 、出口で 335.87K に制御されていればよいことがわかる。図8-1に最適温度パターンと濃度分布の状況を文献の値と比較して図示してある。表8-5では差分の分割数を変化させたときの最適温度パターンについて文献値と比較している。なお、文献では反復修正法を用いた厳密解を計算している。この結果から判るように文献の解と差分近似による解との誤差は濃度で1%程度であり、近似解としては十分である。

2点境界値問題はこの例のほかにも種々の分野で出てくるが、ここで示した方法を適用すれば、EQUATRAN-Mで簡単に解けることが理解できるであろう。

4. 配管網の解析

配管網の問題も、電気回路の問題と同様に典型的な連立方程式解法の問題として取り扱うことができる。電気回路では Ohm の法則が成立するとき、線形問題として容易に解が得られるが、配管網の問題では頭損失(電圧降下に相当)が液流量(電流に相当)に対し非線形な関係にあるため、収束計算が必要となり難解となる。

<例題 8.4: 配管網の解析>⁴⁾

図8-2に示すように、水平面内で正方形形状に位置する点 P_1, P_2, P_3, P_4 と、その中心 P_5 および正方形の3辺の中心点 P_6, P_7 および P_8 という合計8個の地点へ、点 P_1 からある液体原料を供給したい。各点 P_i での所要液量 p_i [m^3/min] は、

$$\begin{aligned} p_1 &= 3.0 & p_2 &= 2.5 & p_3 &= 5.0 & p_4 &= 3.0 \\ p_5 &= 2.0 & p_6 &= 2.0 & p_7 &= 5.0 & p_8 &= 2.5 \end{aligned} \quad (8-23)$$

である。正方形の $1/2$ 辺あたりの頭損失 h_i [m] は、その管を流れる液流量を Q_i [m^3/min] とするとき、高い精度で(8-24)式で近似できることがわかっている注)。

$$h_i = 0.5 |Q_i|^{0.85} Q_i \quad (8-24)$$

この式は、管の曲り部の損失など管摩擦以外の頭損失をも等価的に含んでいる。各管の流量を決定せよ。




図 8-2 配管網のモデル

解) 表8-6にリストと計算結果を示す。図8-2によれば、各管路の流量 $Q_1 \sim Q_9$ の9つの未知数に対して点 $P_2 \sim P_8$ における物質収支が立てられ

表 8-6 例題 8.4 の解

```

1: /* 配管網の解析 */
2: /* (使用記号)-----*/
3: Q(i): 管内放流量 [m3/min]
4: h(i): 正方形 1/2辺あたりの頭損失 [m]
5: p(j): 点 Pj での所要流量 [m3/min]
6: -----*/
7: LOCAL N=9, M=8
8: VAR Q(N), h(N), p(M)
9: /* 頭損失推算 */
10: h = 0.5*(ABS(Q))^0.85/Q
11: /* 閉回路 P1-P5-P8-P7-P4-P1 について */
12: h(2)+h(4)+h(6)-h(8)-2*h(1)=0
13: /* 閉回路 P5-P2-P6-P3-P7-P8-P5 について */
14: h(3)+h(5)+h(7)-h(9)-h(6)-h(4)=0
15: /* 点 P1 のまわりの物質収支 */
16: Q(3) = Q(5) + p(2) /* 点 P2 */
17: Q(7)+Q(9) = p(3) /* 点 P3 */
18: Q(1) = Q(8) + p(4) /* 点 P4 */
19: Q(2) = Q(3)+Q(4)+p(5) /* 点 P5 */
20: Q(5) = Q(7) + p(6) /* 点 P6 */
21: Q(6)+Q(8)=Q(9) + p(7) /* 点 P7 */
22: Q(4) = Q(6) + p(8) /* 点 P8 */
23: /* 入力データ */
24: INPUT p
25: /* 出力データ */
26: OUTPUT Q, p, h
[ 入力値(=) または 初期値(M) ]
p(8) = 3.0 2.5 5.0 3.0
      2.0 2.0 5.0 2.5
Q(3) # 5
Q(8) # 5
Q(1) # 5
[ 計算結果 ]
Q =
1) 8.958363 2) 13.041637 3) 5.877423 4) 5.164214
5) 3.377423 6) 2.664214 7) 1.377423 8) 5.958363
p =
1) 3 2) 2.5 3) 5 4) 3
5) 2 6) 2 7) 5 8) 2.5
h =
1) 28.880069 2) 57.854133 3) 13.241271 4) 10.423842
5) 4.751744 6) 3.063888 7) 0.904158 8) 13.581722
9) 5.409444

```

る。例えば P_7 点に関しては次の物質収支式が成り立つ。

$$Q_6 + Q_8 = Q_9 + p_7 \quad (8-25)$$

次に、図において閉回路 $P_1 P_5 P_8 P_7 P_4 P_1$ の頭損失について各 Q_i に対応するそれを h_i [m] とすると、

$$h_2 + h_4 + h_6 - h_8 - 2h_1 = 0 \quad (8-26)$$

同様な式が閉回路 $P_5 P_2 P_6 P_3 P_7 P_8 P_5$ についても書け(いわゆるエネルギー収支式)。合計 18 の方程式が得られ、 Q_i, h_i の 18 の未知数を解く問題となる。本問は、非線形な (8-24) 式を線形化近似して解くこともできるが、そのまま非線形方程式として EQUATRAN-M により一気に解くことができる。

注) 配管の頭損失の推算には通常次式に示す Williams-Hazen の式が使われるが、本問では簡単のため (8-24) 式を使用している。

$$h = \frac{43.562 \times 10^{14} S Q^{1.85}}{C^{1.85} D^{4.87}}$$

ここで、 h は頭損失 [m]、 S は管延長 [m]、 Q は流量 [m^3/sec]、 C は流速係数 [-]、 D は管径 [m] を表わす。

5. 連続ばりの問題

次に、機械工学分野の問題として、材料力学から連続ばりの問題を取りあげる。この場合も支点が 4 点以上あれば、複数のモーメントに関する式を解く連立方程式解法問題となる。

<例題 8.5: 連続ばりの問題>⁵⁾

図 8-3 に示すように 30m の長さのはり が 4 点で支持され、各々の径間に 1 個ずつの荷重を持っている。支点における曲げモーメントと抗力を求めよ。(はり は長さ方向に均質であるとする。)




図 8-3 連続ばり

解) 一般に図 8-4 に示すような集中荷重を持つばりを考えると、曲げモーメントの式 (8-27) 式が成り立つ。

$$\begin{aligned} & \frac{L_{b1}}{I_{b1}} M_a + 2 \left(\frac{L_{b1}}{I_{b1}} + \frac{L_{b2}}{I_{b2}} \right) M_b + \frac{L_{b2}}{I_{b2}} M_c \\ & + \sum_{r=1}^N \frac{W_{b1r} X_{b1r} (L_{b1}^2 - X_{b1r}^2)}{L_{b1} I_{b1}} \\ & + \sum_{r=1}^N \frac{W_{b2r} X_{b2r} (L_{b2}^2 - X_{b2r}^2)}{L_{b2} I_{b2}} = 0 \quad (8-27) \end{aligned}$$

ここで添字 b は、径間 ABC について式を考える際、中心となる支点が B であることを表わしており、 L_{bi} は支点間の距離 [m]、 I_{bi} は支点間の慣性モーメント [$t \cdot m$]、 W_{bir} は支点間にかかる r




図 8-4 数個の集中荷重をもつ連続ばり




図 8-5 支点Aにおける抗力の計算

表 8-7 例題 8.5 の解

```

1: /* 連続ばりの計算 (材料力学) */
2: LOCAL N = 4 /* 各支点, 荷重数の最大値 */
3: /* 使用記号 */
4: VAR Lb(2), /* 支点間距離 [m] */
5: Lc(2), /* 支点間距離 [m] */
6: Mb, Mc, /* 支点 B, C のまわりの
7: モーメント [t.m] */
8: Ra, Rb, Rc, Rd, /* 支点 A, B, C, D
9: における抗力 [t] */
10: Vb(2,N), /* 荷重 [t] */
11: Vc(2,N), /* 荷重 [t] */
12: Xb(2,N), /* 支点 A, C からの距離 [m] */
13: Xc(2,N), /* 支点 B, D からの距離 [m] */
14: /* 径間 ABC 基準の計算 */
15: 2*( Lb(1)+Lb(2) )#Mb + Lb(2)#Mc
16: +SUM( Vb(1)*Xb(1)*(Lb(1)^2-Xb(1)^2) )/Lb(1)..
17: +SUM( Vb(2)*Xb(2)*(Lb(2)^2-Xb(2)^2) )/Lb(2)..
18: = 0
19: /* 径間 BCD 基準の計算 */
20: Lc(1)#Mb + 2*( Lc(1)+Lc(2) )#Mc
21: +SUM( Vc(1)*Xc(1)*(Lc(1)^2-Xc(1)^2) )/Lc(1)..
22: +SUM( Vc(2)*Xc(2)*(Lc(2)^2-Xc(2)^2) )/Lc(2)..
23: = 0
24: /* 抗力の計算 */
25: - Mb + Lb(1)#Ra = SUM( Vb(1)*(Lb(1)-Xb(1)) )
26: - Mc + Lc(2)#Rd = SUM( Vc(2)*(Lc(2)-Xc(2)) )
27: - Mc + Lb(2)#Rb + (Lb(1)+Lb(2))#Ra
28: = SUM( Vb(2)*Xb(2) )
29: + SUM( Vb(1)*(Lb(1)+Lb(2)-Xb(1)) )
30: - Mb + Lc(1)#Rc + (Lc(1)+Lc(2))#Rd
31: = SUM( Vc(1)*Xc(1) )
32: + SUM( Vc(2)*(Lc(1)+Lc(2)-Xc(2)) )
33: Lc(1) = Lb(2); Vc(1) = Vb(2); Xc(1) = Lb(2) - Xb(2)
34: /* 入力データ */
35: INPUT Lb, Lc(2), Vb, Vc(2), Xb, Xc(2)
36: /* 出力データ */
37: OUTPUT Mb, Mc, Ra, Rb, Rc, Rd
[ 入力値(=) または 初期値(≠) ]
Lb(2) = 10.0 12.0
Lc(2) = 8.0
Vb(4:2) = 5.0 0.0 0.0 0.0
Vc(2)(4) = 7.0 0.0 0.0 0.0
Xb(4:2) = 6.0 0.0 0.0 0.0
Xc(2)(4) = 7.0 0.0 0.0 0.0
Xc(2)(4) = 8.0 0.0 0.0 0.0
Xc(2)(4) = 6.0 0.0 0.0 0.0

```

```

[ 計算結果 ]
Mb = -10.505776
Mc = -7.464934
Ra = 0.449422
Rb = 9.470648
Rc = 7.513047
Rd = 0.566883

```

番目の荷重 [t], X_{bir} は支点から W_{bir} の荷重のある点までの距離 [m], M_a, M_b, M_c はそれぞれ支点 A, B, C における曲げモーメント [t.m] を表わす。

本問では4つの支点で支持されたはりを扱っているので、径間 ABC に関する (8-27) 式を、図 8-3 における径間 BCD にも適用すれば良い (添

字 b を c にすれば良い)。一方、抗力については例えば支点 A における抗力 R_a [t] は図 8-5 に示すように、B 点のすぐ左ではりを切ってその断面のまわりでモーメント収支を考えれば良い。このときの収支式は (8-28) 式のようになる。

$$M_b + L_{b1}R_a - W_{b11}L' = 0 \quad (8-28)$$

同様にして他の支点における抗力を求める。表 8-7 にリストと計算結果を示す。

2 行目の LOCAL 文では、支点間に最大 4 つまでの荷重を載せたケースの検討ができるよう指定を行っている。15~18 行では径間 ABC について、20~23 行では径間 BCD についてそれぞれ曲げモーメント式を記述している。ここで、慣性モーメントの項が消去されているが、はりが長さ方向に均質であり慣性モーメントが一定であるとしたためである。荷重、距離を与え、曲げモーメントを求める場合は線形方程式となり、この問題程度の規模では手計算でも容易に解くことができるが、逆に曲げモーメントと荷重を与えて距離を求める場合は非線形問題となり手計算では難解となるため、EQUATRAN-M の威力が発揮される。

以上、EQUATRAN-M の化学工業を支える周辺技術へのより広い活用法について述べてきた。EQUATRAN-M が自然で書きやすい言語であり、ケミカルエンジニアばかりでなく電気・計測・機械などのエンジニアにとっても有力な武器となることが理解して頂けたと思う。なお、今回は本シリーズのまとめとして、EQUATRAN-M 使用上のノウハウについて述べる予定である。

参考文献

- 1) 須田, 北原: BASIC による電気・電子, pp. 59~62, 共立出版 (1985).
- 2) 日本工業規格: 絞り機構による流量測定方法, Z8762-1969.
- 3) 中谷, 和泉共訳, Liang-Tseng Fan: 最大原理とその応用, コロナ社 (1968).
- 4) 化学工学協会編: 化学工学プログラミング演習, p. 94, 培風館 (1976).
- 5) 太田: 新版材料力学, p. 105, 山海堂 (1966).